

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
				Monarch2090
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	Warteschleife setzt auf 0
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife (vorher 0, nachher 0)
		#define	ZAHL09 rD	Zufallszahl über RND
		#define	TASTE r9	Rote Taste zum Nachstart
		#define	STATUS rA	Wenn Sonderspiele, dann 2, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
0x01	F4C		DISP #4,GELD_PF_EINER	
0x02	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x03	F04		DZHX	
0x04	F4C	start	DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0x05	1E5		MOVI #E,DELTA_D	-20 Pf
0x06	BE5		CALL DeltaSub	
0x07	9FF		CMPI #F,GELD_DM_ZEHNER	Münzspeicher unter 0?
0x08	E00		BRZ start0	dann zurück zum Anfang
0x09	BEE		CALL Warteschleife	
0x0A	FEA		DOT STATUS	Keine Sonderspiele?
0x0B	E10		BRZ StartWalzen	dann direkt Spiel starten
0x0C	F0E		EXRM	
0x0D	F2D		DISP #2,S_EINER	Sonderspiele-Zähler anzeigen
0x0E	BEE		CALL Warteschleife	
0x0F	F0E		EXRM	
0x10	0D5	StartWalzen	MOV GELD_PF_ZEHNER,DELTA_D	
0x11	0E6		MOV GELD_DM_EINER,DELTA_E	
0x12	0F7		MOV GELD_DM_ZEHNER,DELTA_F	
0x13	F50		DISP #5,RO	Walzen darstellen

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x14	BF5		CALL HoleZahl	Linke Walze ermitteln
0x15	1F4	stopL	MOVI #F,LO	König
0x16	1D3		MOVI #D,LU	2 DM
0x17	98D		CMPI #8,ZAHL09	
0x18	D2D		BRC L_fertig	
0x19	1E4		MOVI #E,LO	Krone
0x1A	193		MOVI #9,LU	1 DM
0x1B	97D		CMPI #7,ZAHL09	
0x1C	D2D		BRC L_fertig	
0x1D	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x1E	96D		CMPI #6,ZAHL09	
0x1F	D2D		BRC L_fertig	
0x20	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x21	95D		CMPI #5,ZAHL09	
0x22	D2D		BRC L_fertig	
0x23	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x24	94D		CMPI #4,ZAHL09	
0x25	D2D		BRC L_fertig	
0x26	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x27	93D		CMPI #3,ZAHL09	
0x28	D2D		BRC L_fertig	
0x29	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x2A	91D		CMPI #1,ZAHL09	
0x2B	D2D		BRC L_fertig	
0x2C	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x2D	BF5	L_fertig	CALL HoleZahl	Rechte oder linke Walze ermitteln
0x2E	588		ADDI #8,ZÄHLER	nur einmal Nachstarten ermöglichen
0x2F	D32		BRC stopR	
0x30	929		CMPI #2,TASTE	Taste 2 / Taster H gedrückt?
0x31	E15		BRZ stopL	Dann linke Walze nochmal auswerten
0x32	1F0	stopR	MOVI #F,RO	König
0x33	1D1		MOVI #D,RU	2 DM
0x34	98D		CMPI #8,ZAHL09	
0x35	D47		BRC R_fertig	
0x36	1E0		MOVI #E,RO	Krone
0x37	191		MOVI #9,RU	1 DM
0x38	97D		CMPI #7,ZAHL09	
0x39	D47		BRC R_fertig	
0x3A	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3B	96D		CMPI #6,ZAHL09	
0x3C	D47		BRC R_fertig	
0x3D	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x3E	95D		CMPI #5,ZAHL09	
0x3F	D47		BRC R_fertig	
0x40	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x41	93D		CMPI #3,ZAHL09	
0x42	D47		BRC R_fertig	
0x43	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x44	91D		CMPI #1,ZAHL09	
0x45	D47		BRC R_fertig	
0x46	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x47	BF5	R_fertig	CALL HoleZahl	Mittlere Walze ermitteln
0x48	1F2	stopM	MOVI #F,MI	König
0x49	98D		CMPI #8,ZAHL09	
0x4A	D58		BRC M_fertig	
0x4B	1C2		MOVI #C,MI	1,60 DM
0x4C	97D		CMPI #7,ZAHL09	
0x4D	D58		BRC M_fertig	
0x4E	142		MOVI #4,MI	40 Pf (2x)

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x4F	95D		CMPI #5,ZAHL09	
0x50	D58		BRC M_fertig	
0x51	182		MOVI #8,MI	80 Pf (2x)
0x52	93D		CMPI #3,ZAHL09	
0x53	D58		BRC M_fertig	
0x54	1A2		MOVI #A,MI	1,20 DM (2x)
0x55	91D		CMPI #1,ZAHL09	
0x56	D58		BRC M_fertig	
0x57	1D2		MOVI #D,MI	2 DM (2x)
0x58	05D	M_fertig	MOV DELTA_D,GELD_PF_ZEHNER	
0x59	06E		MOV DELTA_E,GELD_DM_EINER	
0x5A	07F		MOV DELTA_F,GELD_DM_ZEHNER	
0x5B	BEE		CALL Warteschleife	
0x5C	109		MOVI #0,CHANCE2	Gewinnabtastung
0x5D	10B		MOVI #0,GEWINNSYMBOL	
0x5E	BFB		CALL LöscheDelta	
0x5F	840	ChanceA	CMP LO,RO	
0x60	E64		BRZ erste_Chance	
0x61	841		CMP LO,RU	
0x62	E64		BRZ erste_Chance	
0x63	C65		GOTO ChanceB	
0x64	048	erste_Chance	MOV LO,CHANCE1	
0x65	830	ChanceB	CMP LU,RO	
0x66	E6A		BRZ zweite_Chance	
0x67	831		CMP LU,RU	
0x68	E6A		BRZ zweite_Chance	
0x69	C6B		GOTO Mitte	
0x6A	039	zweite_Chance	MOV LU,CHANCE2	
0x6B	9E2	Mitte	CMPI #E,MI	König (Joker) in der Mitte?
0x6C	D76		BRC Joker_ja	
0x6D	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x6E	E70		BRZ gewinn1	
0x6F	C71		GOTO test2	
0x70	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x71	829	test2	CMP MI,CHANCE2	
0x72	E74		BRZ gewinn2	
0x73	C7D		GOTO symbol	
0x74	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x75	C7D		GOTO symbol	
0x76	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x77	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x78	D7A		BRC aaa	
0x79	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x7A	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x7B	D7D		BRC symbol	
0x7C	09B		MOV CHANCE2,GEWINNSYMBOL	bester Gewinn
0x7D	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x7E	E81		BRZ G100	
0x7F	D82		BRC Gewinn120	
0x80	C91		GOTO GewinnBis80Pf	
0x81	1A5	G100	MOVI #A,DELTA_D	1 DM geben (Hex A)
0x82	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x83	E85		BRZ G120	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x84	C86		GOTO Gewinn140	
0x85	1C5	G120	MOVI #C,DELTA_D	1,20 DM geben (Hex C)
0x86	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x87	E89		BRZ G140	
0x88	C8A		GOTO Gewinn160	
0x89	1E5	G140	MOVI #E,DELTA_D	1,40 DM geben (Hex E)
0x8A	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x8B	E8E		BRZ G160	
0x8C	D9B		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x8D	C92		GOTO SoSpTest	
0x8E	105	G160	MOVI #0,DELTA_D	1,60 DM geben (Hex 10)
0x8F	116		MOVI #1,DELTA_E	
0x90	C92		GOTO SoSpTest	
0x91	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x92	90A	SoSpTest	CMPI #0,STATUS	Laufen gerade keine Sonderspiele?
0x93	E9D		BRZ GongTest	
0x94	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x95	D9B		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x96	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x97	E9B		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x98	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x99	D9B		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9A	CA4		GOTO GewinnAnzeige	
0x9B	145	Erhöhung	MOVI #4,DELTA_D	2 DM geben (Hex 14)
0x9C	116		MOVI #1,DELTA_E	
0x9D	906	GongTest	CMPI #0,DELTA_E	
0x9E	DA1		BRC Ton	
0x9F	905		CMPI #0,DELTA_D	Nichts gewonnen?
0xA0	EA4		BRZ GewinnAnzeige	Dann nur anzeigen
0xA1	111	Ton	MOVI #1,RU	
0xA2	FE1		DOT RU	Sound erzeugen, Ausgang 1 auf 1 setzen
0xA3	FEA		DOT STATUS	
0xA4	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA5	F3C		DISP #3,GELD_PF_EINER	
0xA6	BEE		CALL Warteschleife	setzt auch R0 auf 0 wg. EXTRA
0xA7	F02		DISOUT	
0xA8	F0F		EXRA	
0xA9	BE7		CALL DeltaAdd	
0xAA	BFB		CALL LöscheDelta	Geldgewinne löschen
0xAB	90A		CMPI #0,STATUS	Laufen keine Sonderspiele?
0xAC	EC5		BRZ SoGewinn	dann EFE oder FFF testen
0xAD	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xAE	1F5		MOVI #F,DELTA_D	vorher -1 Sonderspiel (akt. Spiel)
0xAF	BE5		CALL DeltaSub	
0xB0	BFB		CALL LöscheDelta	
0xB1	7F2		SUBI #F,MI	Tableau testen: König in der Mitte?
0xB2	DC4		BRC EndeTableau	kein König in der Mitte

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xB3	90E	Tableau	CMPI #0,S_ZEHNER	Pos > 16?
0xB4	DC4		BRC EndeTableau	dann keine Tableaugewinne
0xB5	99D		CMPI #9,S_EINER	Pos > 10?
0xB6	DBE		BRC Pos16	dann Pos16 und Pos15 prüfen
0xB7	EBC		BRZ Gewinn10S	Pos = 10? dann 10 Sonderspiele
0xB8	125		MOVI #2,DELTA_D	Pos < 10? dann mind. 2 Sonderspiele
0xB9	90D		CMPI #0,S_EINER	Pos = 1?
0xBA	DC4		BRC EndeTableau	
0xBB	585	Gewinn20S	ADDI #8,DELTA_D	20 Sonderspiele geben:
				2 Sonderspiele aufstocken auf 10
				und zusätzlich...
0xBC	5A5	Gewinn10S	ADDI #A,DELTA_D	10 Sonderspiele geben (Hex A)
0xBD	FB6		ADC DELTA_E	
0xBE	9ED	Pos16	CMPI #E,S_EINER	Pos > 15?
0xBF	DC2		BRC Gewinn50S	Dann 50 Sonderspiele
0xC0	EC3		BRZ Gewinn2S	Pos = 15? dann 2 Sonderspiele
0xC1	CC4		GOTO EndeTableau	
0xC2	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele geben (Hex 32)
0xC3	125	Gewinn2S	MOVI #2,DELTA_D	
0xC4	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xC5	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >15 Sonderspiele gewonnen?
0xC6	DD1		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xC7	9A5		CMPI #A,DELTA_D	Genau 10 gewonnen?
0xC8	ED1		BRZ SoAnzeige	dann anzeigen
0xC9	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xCA	DCF		BRC ZehnS	
0xCB	ECD		BRZ FünfS	
0xCC	CD1		GOTO SoAnzeige	
0xCD	155	FünfS	MOVI #5,DELTA_D	5 Sonderspiele geben
0xCE	CD0		GOTO SoGewonnen	
0xCF	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben (Hex A)
0xD0	12A	SoGewonnen	MOVI #2,STATUS	Statusregister auf "Sonderspiele" setzen
0xD1	FEA	SoAnzeige	DOT STATUS	Keine Sonderspiele?
0xD2	EE1		BRZ Aufräumen	dann keine Anzeige und aufräumen
0xD3	F0F		EXRA	Sonderspiele-Gewinn anzeigen
0xD4	F2D		DISP #2,S_EINER	
0xD5	BEE		CALL Warteschleife	
0xD6	F0F		EXRA	
0xD7	F02		DISOUT	
0xD8	F0E		EXRM	Sonderspiele-Register einblenden
0xD9	BE7		CALL DeltaAdd	mögl. Sonderspiele-Gewinn addieren
0xDA	90E		CMPI #0,S_ZEHNER	Sonderspiele-Zähler 0?
0xDB	DE0		BRC SoAnzeigeEnde	
0xDC	90D		CMPI #0,S_EINER	
0xDD	DE0		BRC SoAnzeigeEnde	
0xDE	10A		MOVI #0,STATUS	Dann Status zurücksetzen
0xDF	F7A		MAS STATUS	und in Arbeitsregister kopieren
0xE0	F0E	SoAnzeigeEnde	EXRM	Sonderspiele-Register ausblenden

[illegible]

[illegible]

[illegible]

[illegible]

[illegible]

[illegible]