

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comments
				Assemble Instructions
				Monarch2090
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	Warteschleife setzt auf 0
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife (vorher 0, nachher 0)
		#define	ZAHL09 r9	Zufallszahl über Eingänge 1-4
		#define	TASTE r9	Rote Taste zum Nachstart
		#define	STATUS rA	Wenn Sonderspiele, dann 2, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
				DISP #4,GELD_PF_EINER
0x01	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x02	F04		DZHX	
0x03	F4C	start	DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0x04	1E5		MOVI #E,DELTA_D	-20 Pf
0x05	BE6		CALL DeltaSub	
0x06	9FF		CMPI #F,GELD_DM_ZEHNER	Münzspeicher unter 0?
0x07	E00		BRZ start0	dann zurück zum Anfang
0x08	BEF		CALL Warteschleife	
0x09	90A		CMPI #0,STATUS	Keine Sonderspiele?
0x0A	E0F		BRZ StartWalzen	dann direkt Spiel starten
0x0B	F0E		EXRM	
0x0C	F2D		DISP #2,S_EINER	Sonderspiele-Zähler anzeigen
0x0D	BEF		CALL Warteschleife	
0x0E	F0E		EXRM	
0x0F	F50	StartWalzen	DISP #5,RO	Walzen darstellen
0x10	BF6	stopL	CALL HoleZahl	Linke Walze ermitteln
0x11	1F4		MOVI #F,LO	König

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x12	1D3		MOVI #D,LU	2 DM
0x13	989		CMPI #8,ZAHL09	
0x14	D29		BRC L_fertig	
0x15	1E4		MOVI #E,LO	Krone
0x16	193		MOVI #9,LU	1 DM
0x17	979		CMPI #7,ZAHL09	
0x18	D29		BRC L_fertig	
0x19	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x1A	969		CMPI #6,ZAHL09	
0x1B	D29		BRC L_fertig	
0x1C	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x1D	959		CMPI #5,ZAHL09	
0x1E	D29		BRC L_fertig	
0x1F	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x20	949		CMPI #4,ZAHL09	
0x21	D29		BRC L_fertig	
0x22	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x23	939		CMPI #3,ZAHL09	
0x24	D29		BRC L_fertig	
0x25	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x26	919		CMPI #1,ZAHL09	
0x27	D29		BRC L_fertig	
0x28	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x29	145	L_fertig	MOVI #4,DELTA_D	Stoppe und liefere Taste (#E von -20 Pf)
0x2A	FE5		DOT DELTA_D	
0x2B	BEF		CALL Warteschleife	
0x2C	FD9		DIN TASTE	Tastenergebnis
0x2D	FEA		DOT STATUS	Wieder Status anzeigen
0x2E	7F6		SUBI #F,DELTA_E	nur einmal Nachstarten ermöglichen
0x2F	D32		BRC stopR	
0x30	909		CMPI #0,TASTE	Taste gedrückt?
0x31	D10		BRC stopL	Dann linke Walze nochmal starten
0x32	BF6	stopR	CALL HoleZahl	Rechte Walze ermitteln
0x33	1F0		MOVI #F,RO	König
0x34	1D1		MOVI #D,RU	2 DM
0x35	989		CMPI #8,ZAHL09	
0x36	D48		BRC R_fertig	
0x37	1E0		MOVI #E,RO	Krone
0x38	191		MOVI #9,RU	1 DM
0x39	979		CMPI #7,ZAHL09	
0x3A	D48		BRC R_fertig	
0x3B	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3C	969		CMPI #6,ZAHL09	
0x3D	D48		BRC R_fertig	
0x3E	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x3F	959		CMPI #5,ZAHL09	
0x40	D48		BRC R_fertig	
0x41	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x42	939		CMPI #3,ZAHL09	
0x43	D48		BRC R_fertig	
0x44	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x45	919		CMPI #1,ZAHL09	
0x46	D48		BRC R_fertig	
0x47	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x48	BEF	R_fertig	CALL Warteschleife	
0x49	BF6	stopM	CALL HoleZahl	Mittlere Walze ermitteln
0x4A	1F2		MOVI #F,MI	König
0x4B	989		CMPI #8,ZAHL09	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x4C	D5A		BRC M_fertig	
0x4D	1C2		MOVI #C,MI	1,60 DM
0x4E	979		CMPI #7,ZAHL09	
0x4F	D5A		BRC M_fertig	
0x50	142		MOVI #4,MI	40 Pf (2x)
0x51	959		CMPI #5,ZAHL09	
0x52	D5A		BRC M_fertig	
0x53	182		MOVI #8,MI	80 Pf (2x)
0x54	939		CMPI #3,ZAHL09	
0x55	D5A		BRC M_fertig	
0x56	1A2		MOVI #A,MI	1,20 DM (2x)
0x57	919		CMPI #1,ZAHL09	
0x58	D5A		BRC M_fertig	
0x59	1D2		MOVI #D,MI	2 DM (2x)
0x5A	BEF	M_fertig	CALL Warteschleife	
0x5B	F01		NOP	Gewinnabtastung
0x5C	109		MOVI #0,CHANCE2	
0x5D	BFC		CALL LöscheDelta	
0x5E	840	ChanceA	CMP LO,RO	
0x5F	E63		BRZ erste_Chance	
0x60	841		CMP LO,RU	
0x61	E63		BRZ erste_Chance	
0x62	C64		GOTO ChanceB	
0x63	048	erste_Chance	MOV LO,CHANCE1	
0x64	830	ChanceB	CMP LU,RO	
0x65	E69		BRZ zweite_Chance	
0x66	831		CMP LU,RU	
0x67	E69		BRZ zweite_Chance	
0x68	C6A		GOTO Mitte	
0x69	039	zweite_Chance	MOV LU,CHANCE2	
0x6A	9F2	Mitte	CMPI #F,MI	König (Joker) in der Mitte?
0x6B	E6D		BRZ Joker_ja	
0x6C	C75		GOTO Joker_nein	
0x6D	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x6E	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x6F	D71		BRC aaa	
0x70	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x71	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x72	D74		BRC bbb	
0x73	09B		MOV CHANCE2,GEWINNSYMBOL	bester Gewinn
0x74	C7D	bbb	GOTO symbol	
0x75	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x76	E78		BRZ gewinn1	
0x77	C79		GOTO test2	
0x78	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x79	829	test2	CMP MI,CHANCE2	
0x7A	E7C		BRZ gewinn2	
0x7B	C7D		GOTO symbol	
0x7C	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x7D	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x7E	E81		BRZ G100	
0x7F	D82		BRC Gewinn120	
0x80	C91		GOTO GewinnBis80Pf	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment Assemble Instructions
0x81	1A5	G100	MOVI #A,DELTA_D	Hex 10
0x82	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x83	E85		BRZ G120	
0x84	C86		GOTO Gewinn140	
0x85	1C5	G120	MOVI #C,DELTA_D	Hex 12
0x86	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x87	E89		BRZ G140	
0x88	C8A		GOTO Gewinn160	
0x89	1E5	G140	MOVI #E,DELTA_D	Hex 14
0x8A	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x8B	E8E		BRZ G160	
0x8C	D9B		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x8D	C92		GOTO SoSpTest	
0x8E	105	G160	MOVI #0,DELTA_D	Hex 16
0x8F	116		MOVI #1,DELTA_E	Hex 16
0x90	C92		GOTO SoSpTest	
0x91	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x92	90A	SoSpTest	CMPI #0,STATUS	Laufen gerade keine Sonderspiele?
0x93	E9D		BRZ GongTest	
0x94	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x95	D9B		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x96	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x97	E9B		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x98	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x99	D9B		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9A	CA4		GOTO GewinnAnzeige	
0x9B	145	Erhöhung	MOVI #4,DELTA_D	2 DM (Hex 20)
0x9C	116		MOVI #1,DELTA_E	
0x9D	906	GongTest	CMPI #0,DELTA_E	
0x9E	DA1		BRC Ton	
0x9F	905		CMPI #0,DELTA_D	Nichts gewonnen?
0xA0	EA4		BRZ GewinnAnzeige	Dann nur anzeigen
0xA1	111	Ton	MOVI #1,RU	
0xA2	FE1		DOT RU	Sound erzeugen, Ausgang 1 auf 1 setzen
0xA3	FEA		DOT STATUS	
0xA4	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA5	F3C		DISP #3,GELD_PF_EINER	
0xA6	BEF		CALL Warteschleife	setzt auch RO auf 0 wg. EXRA
0xA7	F02		DISOUT	
0xA8	F0F		EXRA	
0xA9	BE8		CALL DeltaAdd	
0xAA	BFC		CALL LöscheDelta	Geldgewinne löschen
0xAB	90A		CMPI #0,STATUS	Laufen keine Sonderspiele?
0xAC	EC7		BRZ SoGewinn	dann EFE oder FFF testen
0xAD	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xAE	1F5		MOVI #F,DELTA_D	vorher -1 Sonderspiel (akt. Spiel)

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xAF	BE6		CALL DeltaSub	
0xB0	BFC		CALL LöscheDelta	
0xB1	7F2		SUBI #F,MI	Tableau testen: König in der Mitte?
0xB2	DC6		BRC EndeTableau	kein König in der Mitte
0xB3	90E	Tableau	CMPI #0,S_ZEHNER	Pos > 16?
0xB4	DC6		BRC EndeTableau	dann keine Tableaugewinne
0xB5	99D		CMPI #9,S_EINER	Pos > 10?
0xB6	DC0		BRC Pos16	dann Pos16 und Pos15 prüfen
0xB7	EBE		BRZ Gewinn10S	Pos = 10? dann 10 Sonderspiele
0xB8	125		MOVI #2,DELTA_D	Pos < 10? dann mind. 2 Sonderspiele
0xB9	90D		CMPI #0,S_EINER	Pos = 1?
0xBA	DC6		BRC EndeTableau	
0xBB	116	Gewinn20S	MOVI #1,DELTA_E	20 Sonderspiele (Hex 14)
0xBC	145		MOVI #4,DELTA_D	
0xBD	CC6		GOTO EndeTableau	
0xBE	1A5	Gewinn10S	MOVI #A,DELTA_D	10 Sonderspiele (Hex A)
0xBF	CC6		GOTO EndeTableau	
0xC0	9ED	Pos16	CMPI #E,S_EINER	Pos > 15?
0xC1	DC4		BRC Gewinn50S	Dann 50 Sonderspiele
0xC2	EC5		BRZ Gewinn2S	Pos = 15? dann 2 Sonderspiele
0xC3	CC6		GOTO EndeTableau	
0xC4	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele (Hex 32)
0xC5	125	Gewinn2S	MOVI #2,DELTA_D	
0xC6	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xC7	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >15 Sonderspiele gewonnen?
0xC8	DD3		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xC9	9A5		CMPI #A,DELTA_D	Genau 10 gewonnen?
0xCA	ED3		BRZ SoAnzeige	dann anzeigen
0xCB	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xCC	DD1		BRC ZehnS	
0xCD	ECF		BRZ FünfS	
0xCE	CD3		GOTO SoAnzeige	
0xCF	155	FünfS	MOVI #5,DELTA_D	5 Sonderspiele geben
0xD0	CD2		GOTO SoGewonnen	
0xD1	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben
0xD2	12A	SoGewonnen	MOVI #2,STATUS	Statusregister auf "Sonderspiele" setzen
0xD3	90A	SoAnzeige	CMPI #0,STATUS	Keine Sonderspiele?
0xD4	EE3		BRZ Aufräumen	dann keine Anzeige und aufräumen
0xD5	F0F		EXRA	Sonderspiele-Gewinn anzeigen
0xD6	F2D		DISP #2,S_EINER	
0xD7	BEF		CALL Warteschleife	
0xD8	F0F		EXRA	
0xD9	F02		DISOUT	DISOUT nötig?
0xDA	F0E		EXRM	Sonderspiele-Register einblenden
0xDB	BE8		CALL DeltaAdd	mögl. Sonderspiele-Gewinn addieren
0xDC	90E		CMPI #0,S_ZEHNER	Sonderspiele-Zähler 0?





[illegible]



[illegible]

[illegible]

[illegible]

[illegible]

[illegible]

