

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
				<b>Monarch2090</b>
				##### Register use #####
		#define	LO r4	Walze links oben
		#define	LU r3	Walze links unten
		#define	MI r2	Walze in der Mitte
		#define	RO r0	Walze rechts oben
		#define	RU r1	Walze rechts unten
		#define	GELD_PF_EINER rC	Münzspeicher 1 Pf (immer 0)
		#define	GELD_PF_ZEHNER rD	Münzspeicher 10 Pf
		#define	GELD_DM_EINER rE	Münzspeicher 1 DM
		#define	GELD_DM_ZEHNER rF	Münzspeicher 10 DM
		#define	S_EINER rD	Sonderspiele Einer (im Speicherregister)
		#define	S_ZEHNER rE	Sonderspiele Zehner (im Speicherregister)
		#define	S_HUNDERTER rF	Sonderspiele Hunderter (im SR, immer 0)
		#define	DELTA_C r4	auf 0 setzen nach Walzenlauf wg. Anzeige
		#define	DELTA_D r5	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_E r6	Gewinn/Abzug Geld/Sonderspiele
		#define	DELTA_F r7	Gewinn/Abzug Geld/Sonderspiele
		#define	ZÄHLER r8	für Warteschleife
		#define	ZAHL09 r9	Zufallszahl über Eingänge 1-4
		#define	TASTE r9	nicht verwendet, optional
		#define	STATUS rA	Wenn Sonderspiele, dann 1, sonst 0
		#define	CHANCE1 r8	Symbol links und rechts gleich I
		#define	CHANCE2 r9	Symbol links und rechts gleich II
		#define	GEWINNSYMBOL rB	erzieltes Gewinnsymbol
				##### Program #####
0x00	F08	start0	CLEAR	
0x01	F4C		DISP #4,GELD_PF_EINER	
0x02	FFE		KIN GELD_DM_EINER	Geld einwerfen
0x03	F04		DZHX	
0x04	1E5	start	MOVI #E,DELTA_D	-20 Pf
0x05	BEB		CALL DeltaSub	
0x06	9FF		CMPI #F,rF	RF=F? Dann Speicher leer
0x07	E00		BRZ start0	
0x08	90A		CMPI #0,STATUS	Keine Sonderspiele?
0x09	E16		BRZ start2	Dann Zähler nicht anzeigen
0x0A	F0E		EXRM	
0x0B	F2D		DISP #2,S_EINER	
0x0C	BF4		CALL Anzeige	Sonderspiele anzeigen
0x0D	1F5		MOVI #F,DELTA_D	-1 Spiel
0x0E	BEB		CALL DeltaSub	
0x0F	9FF		CMPI #F,rF	Sonderspiele nach Abzug unter 0?
0x10	E12		BRZ LöscheStatus	dann Status auf 0 setzen (keine Sonderspiele mehr)
0x11	C14		GOTO start1	
0x12	10A	LöscheStatus	MOVI #0,STATUS	

#Mem Addr	#Ma- chine Code	#Branch/Call Target	#Mnemonic	Assemble Instructions
0x13	F7A		MAS STATUS	Status in Speicherregister kopieren
0x14	BF4	start1	CALL Anzeige	
0x15	F0E		EXRM	
0x16	F4C	start2	DISP #4,GELD_PF_EINER	
0x17	BF4		CALL Anzeige	
0x18	F50		DISP #5,RO	Walzen darstellen
0x19	FD9	stopL	DIN ZAHL09	Linke Walze ermitteln
0x1A	1F4		MOVI #F,LO	König
0x1B	1D3		MOVI #D,LU	2 DM
0x1C	989		CMPI #8,ZAHL09	
0x1D	D32		BRC L_fertig	
0x1E	1E4		MOVI #E,LO	Krone
0x1F	193		MOVI #9,LU	1 DM
0x20	979		CMPI #7,ZAHL09	
0x21	D32		BRC L_fertig	
0x22	1B4		MOVI #B,LO	1,40 DM (1 DM LU schon gesetzt)
0x23	969		CMPI #6,ZAHL09	
0x24	D32		BRC L_fertig	
0x25	184		MOVI #8,LO	80 Pf (1 DM LU schon gesetzt)
0x26	959		CMPI #5,ZAHL09	
0x27	D32		BRC L_fertig	
0x28	143		MOVI #4,LU	40 Pf (80 Pf LO schon gesetzt)
0x29	949		CMPI #4,ZAHL09	
0x2A	D32		BRC L_fertig	
0x2B	1B4		MOVI #B,LO	1,40 DM (40 Pf RU schon gesetzt)
0x2C	939		CMPI #3,ZAHL09	
0x2D	D32		BRC L_fertig	
0x2E	1A4		MOVI #A,LO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x2F	919		CMPI #1,ZAHL09	
0x30	D32		BRC L_fertig	
0x31	1C4		MOVI #C,LO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x32	BF4	L_fertig	CALL Anzeige	
0x33	FD9	stopR	DIN ZAHL09	Rechte Walze ermitteln
0x34	1F0		MOVI #F,RO	König
0x35	1D1		MOVI #D,RU	2 DM
0x36	989		CMPI #8,ZAHL09	
0x37	D49		BRC R_fertig	
0x38	1E0		MOVI #E,RO	Krone
0x39	191		MOVI #9,RU	1 DM
0x3A	979		CMPI #7,ZAHL09	
0x3B	D49		BRC R_fertig	
0x3C	180		MOVI #8,RO	80 Pf (1 DM RU schon gesetzt)
0x3D	969		CMPI #6,ZAHL09	
0x3E	D49		BRC R_fertig	
0x3F	1B1		MOVI #B,RU	1,40 DM (80 Pf RO schon gesetzt)
0x40	959		CMPI #5,ZAHL09	
0x41	D49		BRC R_fertig	
0x42	141		MOVI #4,RU	40 Pf (80 Pf RO schon gesetzt) (2x)
0x43	939		CMPI #3,ZAHL09	
0x44	D49		BRC R_fertig	
0x45	1C0		MOVI #C,RO	1,60 DM (40 Pf RU schon gesetzt) (2x)
0x46	919		CMPI #1,ZAHL09	
0x47	D49		BRC R_fertig	
0x48	1A0		MOVI #A,RO	1,20 DM (40 Pf RU schon gesetzt) (2x)
0x49	BF4	R_fertig	CALL Anzeige	
0x4A	FD9	stopM	DIN ZAHL09	Mittlere Walze ermitteln

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x4B	1F2		MOVI #F,MI	König
0x4C	989		CMPI #8,ZAHL09	
0x4D	D5B		BRC M_fertig	
0x4E	1C2		MOVI #C,MI	1,60 DM
0x4F	979		CMPI #7,ZAHL09	
0x50	D5B		BRC M_fertig	
0x51	142		MOVI #4,MI	40 Pf (2x)
0x52	959		CMPI #5,ZAHL09	
0x53	D5B		BRC M_fertig	
0x54	182		MOVI #8,MI	80 Pf (2x)
0x55	939		CMPI #3,ZAHL09	
0x56	D5B		BRC M_fertig	
0x57	1A2		MOVI #A,MI	1,20 DM (2x)
0x58	919		CMPI #1,ZAHL09	
0x59	D5B		BRC M_fertig	
0x5A	1D2		MOVI #D,MI	2 DM (2x)
0x5B	BF4	M_fertig	CALL Anzeige	
0x5C	108		MOVI #0,CHANCE1	Gewinnabtastung
0x5D	109		MOVI #0,CHANCE2	
0x5E	10B		MOVI #0,GEWINNSYMBOL	
0x5F	BFA		CALL LöscheDelta	
0x60	840	ChanceA	CMP LO,RO	
0x61	E65		BRZ erste_Chance	
0x62	841		CMP LO,RU	
0x63	E65		BRZ erste_Chance	
0x64	C66		GOTO ChanceB	
0x65	048	erste_Chance	MOV LO,CHANCE1	
0x66	830	ChanceB	CMP LU,RO	
0x67	E6B		BRZ zweite_Chance	
0x68	831		CMP LU,RU	
0x69	E6B		BRZ zweite_Chance	
0x6A	C6C		GOTO Mitte	
0x6B	039	zweite_Chance	MOV LU,CHANCE2	
0x6C	9F2	Mitte	CMPI #F,MI	König (Joker) in der Mitte?
0x6D	E6F		BRZ Joker_ja	
0x6E	C77		GOTO Joker_nein	
0x6F	12B	Joker_ja	MOVI #2,GEWINNSYMBOL	Mindestens 20 Pf
0x70	88B		CMP CHANCE1,GEWINNSYMBOL	Höchste Chance = Gewinn
0x71	D73		BRC aaa	
0x72	08B		MOV CHANCE1,GEWINNSYMBOL	besserer Gewinn
0x73	89B	aaa	CMP CHANCE2,GEWINNSYMBOL	
0x74	D76		BRC bbb	
0x75	09B		MOV CHANCE2,GEWINNSYMBOL	bestser Gewinn
0x76	C7E	bbb	GOTO symbol	
0x77	828	Joker_nein	CMP MI,CHANCE1	Normale Gewinnmöglichkeit
0x78	E7A		BRZ gewinn1	
0x79	C7B		GOTO test2	
0x7A	08B	gewinn1	MOV CHANCE1,GEWINNSYMBOL	
0x7B	829	test2	CMP MI,CHANCE2	
0x7C	E7E		BRZ gewinn2	
0x7D	C7F		GOTO symbol	
0x7E	09B	gewinn2	MOV CHANCE2,GEWINNSYMBOL	
0x7F	99B	symbol	CMPI #9,GEWINNSYMBOL	1 DM?
0x80	E83		BRZ G100	

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0x81	D84		BRC Gewinn120	
0x82	C93		GOTO GewinnBis80Pf	
0x83	1A5	G100	MOVI #A,DELTA_D	Hex 10
0x84	9AB	Gewinn120	CMPI #A,GEWINNSYMBOL	1,20 DM?
0x85	E87		BRZ G120	
0x86	C88		GOTO Gewinn140	
0x87	1C5	G120	MOVI #C,DELTA_D	Hex 12
0x88	9BB	Gewinn140	CMPI #B,GEWINNSYMBOL	1,40 DM?
0x89	E8B		BRZ G140	
0x8A	C8C		GOTO Gewinn160	
0x8B	1E5	G140	MOVI #E,DELTA_D	Hex 14
0x8C	9CB	Gewinn160	CMPI #C,GEWINNSYMBOL	1,60 DM?
0x8D	E90		BRZ G160	
0x8E	D9E		BRC Erhöhung	Mehr als 1,60 DM, dann 2 DM geben (Gewinnkombinationen DDD, EFE, FFF)
0x8F	C94		GOTO SoSpTest	
0x90	105	G160	MOVI #0,DELTA_D	Hex 16
0x91	116		MOVI #1,DELTA_E	Hex 16
0x92	C94		GOTO SoSpTest	
0x93	0B5	GewinnBis80Pf	MOV GEWINNSYMBOL,DELTA_D	00 Pf, 20 Pf, 40 Pf, 80 Pf (Hex = Dez)
0x94	91A	SoSpTest	CMPI #1,STATUS	Laufen gerade Sonderspiele?
0x95	E97		BRZ Test2DM	dann auf kariertes Feld oder Gewinn testen
0x96	CA1		GOTO GongTest	
0x97	90B	Test2DM	CMPI #0,GEWINNSYMBOL	Irgendeinen Gewinn erzielt?
0x98	D9E		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x99	942	TestKaro	CMPI #4,MI	Mitte 40 Pf?
0x9A	E9E		BRZ Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9B	9C2		CMPI #C,MI	Mitte 2 DM oder König?
0x9C	D9E		BRC Erhöhung	dann Gewinn auf 2 DM erhöhen
0x9D	CA6		GOTO GewinnAnzeige	
0x9E	145	Erhöhung	MOVI #4,DELTA_D	2 DM (Hex 20)
0x9F	116		MOVI #1,DELTA_E	
0xA0	CA3		GOTO Ton	direkt Ton ausgeben
0xA1	905	GongTest	CMPI #0,DELTA_D	Nichts gewonnen?
0xA2	EA6		BRZ GewinnAnzeige	Dann anzeigen
0xA3	1F9	Ton	MOVI #F,TASTE	
0xA4	FE9		DOT TASTE	Sound erzeugen, Ausgang 1 auf 1 setzen
0xA5	FEC		DOT GELD_PF_EINER	Ausgang 1 auf 0 setzen
0xA6	F0F	GewinnAnzeige	EXRA	Gewinn anzeigen
0xA7	10C		MOVI #0,GELD_PF_EINER	LO löschen
0xA8	F3C		DISP #3,GELD_PF_EINER	
0xA9	BF4		CALL Anzeige	
0xAA	F0F		EXRA	
0xAB	F4C		DISP #4,GELD_PF_EINER	Münzspeicher anzeigen
0xAC	BED		CALL DeltaAdd	
0xAD	BF4		CALL Anzeige	
0xAE	F02		DISOUT	
0xAF	BFA		CALL LöscheDelta	Geldgewinne löschen
0xB0	91A	Tableau	CMPI #1,STATUS	Laufen gerade Sonderspiele?

#Mem Addr	#Machine Code	#Branch/Call Target	#Mnemonic	#Comment
				Assemble Instructions
0xB1	EB3		BRZ TableauTest	dann Gewinn tableau testen
0xB2	CD3		GOTO SoGewinn	
0xB3	F0E	TableauTest	EXRM	Sonderspiele-Register einblenden
0xB4	9F2		CMPI #F,MI	Tableau testen: König in der Mitte?
0xB5	EB7		BRZ Tableau2St	
0xB6	CD2		GOTO EndeTableau	kein König in der Mitte
0xB7	91E	Tableau2St	CMPI #1,S_ZEHNER	Pos 16?
0xB8	EBA		BRZ Tableau16	
0xB9	CC0		GOTO Tableau1St	
0xBA	90D	Tableau16	CMPI #0,S_EINER	
0xBB	EBD		BRZ Gewinn50S	
0xBC	CC0		GOTO Tableau1St	
0xBD	136	Gewinn50S	MOVI #3,DELTA_E	50 Sonderspiele (Hex 32)
0xBE	125		MOVI #2,DELTA_D	
0xBF	CD2		GOTO EndeTableau	
0xC0	90E	Tableau1St	CMPI #0,S_ZEHNER	Positionen 1-15?
0xC1	DD2		BRC EndeTableau	
0xC2	99D		CMPI #9,S_EINER	Pos 1-9
0xC3	DCA		BRC Tableau10	
0xC4	125		MOVI #2,DELTA_D	Mind 2 Sonderspiele geben
0xC5	91D		CMPI #1,S_EINER	Pos1?
0xC6	DD2		BRC EndeTableau	
0xC7	145	Gewinn20S	MOVI #4,DELTA_D	20 Sonderspiele (Hex 14)
0xC8	116		MOVI #1,DELTA_E	
0xC9	CD2		GOTO EndeTableau	
0xCA	9AD	Tableau10	CMPI #A,S_EINER	Pos 10?
0xCB	ECD		BRZ Gewinn10S	
0xCC	CCE		GOTO Tableau14	
0xCD	1A5	Gewinn10S	MOVI #A,DELTA_D	
0xCE	9FD	Tableau14	CMPI #F,S_EINER	Pos 15?
0xCF	ED1		BRZ Gewinn2S	
0xD0	CD2		GOTO EndeTableau	
0xD1	125	Gewinn2S	MOVI #2,DELTA_D	
0xD2	F0E	EndeTableau	EXRM	Sonderspiele-Register ausblenden
0xD3	906	SoGewinn	CMPI #0,DELTA_E	Über Tableau >=10 Sonderspiele gewonnen?
0xD4	DDF		BRC SoAnzeige	dann anzeigen, da mehr als über EFE/FFF
0xD5	9EB	TestEFE	CMPI #E,GEWINNSYMBOL	Sonderspiele über EFE/FFF gewonnen?
0xD6	ED8		BRZ Fünfs	
0xD7	CDA		GOTO TestFFF	
0xD8	155	Fünfs	MOVI #5,DELTA_D	5 Sonderspiele geben
0xD9	CDE		GOTO SoGewonnen	
0xDA	9FB	TestFFF	CMPI #F,GEWINNSYMBOL	
0xDB	EDD		BRZ ZehnS	
0xDC	CDF		GOTO SoAnzeige	
0xDD	1A5	ZehnS	MOVI #A,DELTA_D	10 Sonderspiele geben
0xDE	11A	SoGewonnen	MOVI #1,STATUS	Statusregister auf "Sonderspiele" setzen
0xDF	90A	SoAnzeige	CMPI #0,STATUS	Keine Sonderspiele?
0xE0	EE9		BRZ Aufräumen	dann keine Anzeige und aufräumen

[illegible]